```
Maxima 5.13.0 http://maxima.sourceforge.net
Using Lisp GNU Common Lisp (GCL) GCL 2.6.7 (aka GCL)
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
This is a development version of Maxima. The function bug_report()
provides bug reporting information.
```

Find Gaussian quadrature of order $2n-1$ on $[a,b]$ with weight function w.

Note: If the weight function is complicated, then the integrals in the Gram-Schmidt algorithm may not have closed form solutions. In this case replace integrate by quad_qags in suitable places to compute the coeficients numerically.

```
(%i1) kill(all)$
      n:3;
```

(%o1) 3

```
(%i2) a:0;
      b:1;
      w(x):=1;
```

(%o2) 0

(%o3) 1

(%o4) $w(x):=1$

```
(%i5) for k:0 thru n do (
         q[k]:x^k,
         for j:0 thru k-1 do (
            q[k]:q[k]-q[j]*integrate(q[k]*q[j]*w(x),x,a,b)
         ),
         q[k]:fullratsimp(q[k]/sqrt(integrate(q[k]^2*w(x),x,a,b)))
      )$
(%i6) p:q[n];
```

(%o6) $20\sqrt{7}\,x^3-30\sqrt{7}\,x^2+12\sqrt{7}\,x-\sqrt{7}$

```
(%i7) ctmp:solve(p=0,x), solveexplicit:true;
```

(%o7) $\left[x=-\frac{\sqrt{15}-5}{10},x=\frac{\sqrt{15}+5}{10},x=\frac{1}{2}\right]$

```
(%i8) c:fullratsimp(create_list(subst(ctmp[i],x),i,1,n));
```

(%o8) $\left[-\frac{\sqrt{15}-5}{10},\frac{\sqrt{15}+5}{10},\frac{1}{2}\right]$

```
(%i9) V:vandermonde_matrix(c);
```

(%o9) $$\begin{pmatrix} 1 & -\frac{\sqrt{15}-5}{10} & \frac{\left(\sqrt{15}-5\right)^2}{100} \\ 1 & \frac{\sqrt{15}+5}{10} & \frac{\left(\sqrt{15}+5\right)^2}{100} \\ 1 & \frac{1}{2} & \frac{1}{4} \end{pmatrix}$$

```
(%i10) Y:makelist(integrate(x^k*w(x),x,a,b),k,0,n-1);
```

(%o10) $\left[1, \frac{1}{2}, \frac{1}{3}\right]$

```
(%i11) B:fullratsimp(linsolve_by_lu(transpose(V),Y)[1]);
```

(%o11) $\begin{pmatrix} \frac{5}{18} \\ \frac{5}{18} \\ \frac{4}{9} \end{pmatrix}$

This is the Gaussian quadrature formula such that

$$F(f) \approx \int_a^b f(x)w(x)\mathrm{dx}$$

```
(%i12) define(F(f),map(f,c).B);
```

(%o12) $F(f) := \frac{5\, f\left(\frac{\sqrt{15}+5}{10}\right)}{18} + \frac{5\, f\left(-\frac{\sqrt{15}-5}{10}\right)}{18} + \frac{4\, f\left(\frac{1}{2}\right)}{9}$

Testing this with polynomials shows the quadrature is exact for polynomals of degree $2n-1$ and less.

```
(%i13) for k:0 thru 2*n do
          print('integrate(x^k*w(x),x,a,b)=
              fullratsimp(integrate(x^k*w(x),x,a,b)),'" and ",
          'F(x^k)=fullratsimp(F(lambda([x],x^k))))$
```

$1 = 1$ and $F(1) = 1$

$\int_0^1 x \,\mathrm{d}\, x = \frac{1}{2}$ and $F(x) = \frac{1}{2}$

$\int_0^1 x^2 \,\mathrm{d}\, x = \frac{1}{3}$ and $F\left(x^2\right) = \frac{1}{3}$

$\int_0^1 x^3 \,\mathrm{d}\, x = \frac{1}{4}$ and $F\left(x^3\right) = \frac{1}{4}$

$\int_0^1 x^4 \,\mathrm{d}\, x = \frac{1}{5}$ and $F\left(x^4\right) = \frac{1}{5}$

$\int_0^1 x^5 \,\mathrm{d}\, x = \frac{1}{6}$ and $F\left(x^5\right) = \frac{1}{6}$

$\int_0^1 x^6 \,\mathrm{d}\, x = \frac{1}{7}$ and $F\left(x^6\right) = \frac{57}{400}$

```
(%i14)
```